

FM-8用インテリジェント端末プログラム

入江, 啓一
九州大学大型計算機センター業務掛

末永, 正
九州大学大型計算機センター研究開発部

松延, 尚
富士通SE

<https://doi.org/10.15017/1470269>

出版情報 : 九州大学大型計算機センター広報. 15 (5), pp.515-529, 1982-11-20. 九州大学大型計算機センター
バージョン :
権利関係 :



FM-8用インテリジェント端末プログラム

入江 啓一* 末永 正** 松延 尚***

1. はじめに

最近、パーソナルコンピュータ（以下、パソコンと略す）の普及はめざましく、大学の研究室にもかなりの数が設置されている。利用者の中にはパソコンをスタンドアロンで使うだけでなく、ホストコンピュータ（以下、単にホストと略す）と結びTSS端末として使用したいという要望がかなりある。本センターではこの現状を踏まえ、去る7月9日にインテリジェント端末同好会を結成した。これは、利用者間の情報交換を円滑にし、パソコンをインテリジェント端末にするためのノウハウを公開しようことをめざしたものである。

ところで、本センターでもFUJITSU MICRO8（通称 FM-8）を購入し、そのインテリジェント端末プログラムの作成を試行した。FM-8 向けのこの種のプログラムでは、すでに名大大型計算機センター（以下、名大センターと略す）の平松敏祐氏によるもの〔1〕のはかいくつか実現している。平松氏のプログラムはおもにファイル転送を可能にしたもので、最近はやりのスクリーン編集機能を有していない。そこで、本センターでは、作成者の了解を得てその機能強化を図った。

以下に、新しい端末制御プログラム（TCP/C V04/L04）を紹介する。

2. 機能追加と変更点

名大センターと本センターでは通信制御方式が多少異っており、本センターではホストから送られてくるメッセージの終りにNULLコード（▼00▼）が付加されるが、名大センターでは、固有の事情によりそれを付加しないようにしている。このため名大センター用に作成されたプログラムをそのまま本センターで実行させても正常に動作しない。本センターの場合、NULLコードを意識する必要がある。

また、原プログラムでは、ホストとパソコンの交信時のメッセージをプリンターにエコーバックする機能を有していたが、エコーバック有無の切り替え動作がうまく働かないことや、プリンターの印字速度が非常に遅く通信速度を低下させるのでこの機能は削除した。

追加した機能は次の2つである。

- (1) スクリーン操作
- (2) 高速の双方向ファイル転送

3. TCP コマンド

本プログラムでは、先頭1文字が!で始まる文字列はすべて端末自身のコマンド（TCPコマンドと呼ぶ）と解釈される。TCPコマンドには次の9種類がある。

* 九州大学大型計算機センター業務掛
** 同 研究開発部
*** 富士通 SE

- ! ホストへ割込信号を送出し、実行中の処理を一時中断する。
- !C CRT画面のハードコピーをプリンターに出力する。
- !E LOGOFFコマンドをホストへ送出する。
- !M オーディオカセットレコーダーのモーターのON/OFFを制御する。
- !H または !? TCPコマンド一覧とPFキーの機能を表示する。
- !L LOGONコマンドをホストへ送出して、TSSセッションを開設する。
- !S 端末制御プログラムを一時中断して、BASICのコマンドモードにする。BASICのCONTコマンドを入力すれば再開する。
- !F このコマンドは次の6つの機能を有し、メニュー形式で表示される。
 - ① ホストのデータセットをパソコンへ転送する（新規作成）。
 - ② ホストのデータセットをパソコンへ転送する（既存ファイルへ追加）。
 - ③ パソコンのファイルをホストへ転送する。
 - ④ ファイルの内容をCRTへ表示する。
 - ⑤ ファイルの内容をプリンターへ出力する。
 - ⑥ ファイルのクローズ、または!Fを終了する。
- !TR ファイル転送のみを高速に処理する。

4. PFキー

- PF1 ホストへ割込信号を送出する。
- PF10 本プログラムの実行を終了する。

5. スクリーン操作

① カーソル表示

F-BASICでは、プログラム実行中はカーソルが表示されない。このため、本プログラムでは次のようにしてカーソルを表示している。

- (イ) 画面上のカーソル位置の1文字を配列変数に
 組み込む(図1)。
- (ロ) Rを反転させて表示する。
- (ハ) Rを元に戻して表示する。
- (ニ) (イ)～(ハ)を一定間隔で繰り返す。




図1.

② コマンドの作成

本プログラムでは、画面上の任意の行を入力行にすることができる。入力行としての認識は、その行に何らかの変更がなされた時であり、変更した行を認識するために入力行識別子(図2)を持っている。入力行識別子の値が1であればその行を入力行と認識する。従って画面全体




図2.

(25行, 1行80文字)を入力行にすることも可能であるが, 文字変数の最大長が255バイトとなっているため, 現在のところ3行までに制限している。

③ スクリーン制御キー

スクリーン操作を行うための特殊キーとその意味を以下に示す。

キー	コード(16進)	動作
E L	05	カーソル位置以降削除
B S	08	カーソルの直前の文字削除
T A B	09	カーソルを左端へ移動
H O M E	0B	カーソルをホームポジションへ移動
C L E A R	0C	画面をクリア
R E T U R N	0D	入力行を送信する
D U P	12	カーソル位置の上の文字を複写する
I N S	13	カーソル位置へ文字挿入
→	1C	カーソルを右へ移動
←	1D	カーソルを左へ移動
↑	1E	カーソルを上へ移動
↓	1F	カーソルを下へ移動
D E L	7F	カーソル位置の文字削除
未定義キー	00 ~ 1F	内の上記以外の文字 警告音

6. ソースプログラム

新しい端末制御プログラム(TCP/C V04/L04)を付録1に示す。本プログラムは, 次のような構成を取っている。

10 ~ 390	初期設定
400 ~ 1460	スクリーンエディタ
1470 ~ 1600	コマンド選択および送信
1610 ~ 1700	ブレーク処理
1710 ~ 1720	!Cコマンド(画面のハードコピー)処理
1730 ~ 1740	!Eコマンド(LOGOFF)処理
1750 ~ 1760	!Lコマンド(LOGON)処理
1770 ~ 1790	!Sコマンド(実行を一時中断する)処理
1800 ~ 1840	TERMINALコマンドの定義
1850 ~ 1910	!Hコマンド(TCPコマンド一覧とPFキーの機能を表示)処理
1920 ~ 2670	!Fコマンド処理
2680 ~ 2740	ホストからのメッセージ表示
2750 ~ 2780	エラー処理
2790 ~ 3250	!TRコマンド(高速双方向ファイル転送)処理

7. 使用方法

本端末制御プログラムの使用方法を簡単に説明する。まずプログラムをLOADし、RUNコマンドを入力するとLOGON要求のメッセージが表示されてカーソルが点滅し始める。そこで“!L”を入力する（LOGONコマンド自身を入力してもよい。ただしこのときは、タイムアウト検出機能の抑止と論理行サイズの変更処理として、▼READY▼が表示された後に▼TERM_NOTIMEOUT_LINESIZE(225)▼を入力する必要がある）とTSSセッションが開設される。以後は通常のTSS端末と同様であるが、アテンションがPF1であること、およびコマンド入力モードにおいて、先頭文字が“!”であればTCPコマンドとして解釈されることに注意して頂きたい。

8. 受信専用プログラム

端末制御プログラムはホスト上に保存しており、そのデータセット名は▼F0017.TERMProg.DATA(FM8TCP)▼である。そこで、このデータセットをFM-8の補助記憶装置に転送する受信専用のプログラムを用意した。利用者は、以下に示す受信専用プログラムを入力し、BASICのRUNコマンドでそれを実行する。“PLEASE LOGON”のメッセージが表示された後、LOGONコマンドを入力し、TSSセッションを開設する。READYを確認して“FMCOPY”を入力するとファイル転送が開始する。受信終了後、LOGOFFコマンドを入力し、TSSセッションが終了したことを確認してSTOP キーを押下する。これで、本来の端末制御プログラムを利用できるわけであるが、この受信専用プログラムは、補助記憶装置をミニロッピディスクとし、ファイル名をTCP/V4L4と定義している。変更したければ、行番号150を書きかえるとよい。また、受信専用プログラムで作成したファイルは、アスキー形式であるため、バイナリー形式で再セーブする方がローディング時間が短くて済む。さらに、行番号1760を自分の課題番号とパスワードに修正してSAVEしておくで“!L”コマンドによってLOGON手続きが簡単になる。

受信専用プログラムを以下に示す。

```

10 COLOR 4,0: CLEAR 1000: WIDTH 80,25: CONSOLE 0,25,0,0: KEY(1) ON: KEY(2) ON: ON KEY(
1) GOSUB 270: ON KEY(2) GOSUB 260: ON ERROR GOTO 250
20 OPEN "I", #1, "COM0: (S7E1)": OPEN "O", #2, "COM0: (S7E1)"
30 EXEC &HD748: LOCATE 0,0: PRINT "PLEASE LOGON": CS$=""
40 COLOR 4: A$=INKEY$: IF A$="" THEN GOTO 80 ELSE CD=ASC(A$)
50 IF CD=8 THEN CS$=LEFT$(CS$,LEN(CS$)-1): LOCATE POS(0)-1,CSRLIN: PRINT " ";: LOCA
TE POS(0)-1,CSRLIN: GOTO 40
60 PRINT A$;: IF CD<>13 THEN CS$=CS$+A$: GOTO 40
70 IF CS$="FMCOPY" THEN CS$="": GOTO 90 ELSE PRINT #2,CS$:CS$=""
80 IF LOF(1)=0 THEN GOTO 40 ELSE PRINT INPUT$(LOF(1),#1);: GOTO 40
90 RQ$=CHR$(2)+CHR$(&H1E)+CHR$(3): EDRD$=CHR$(2)+CHR$(&H1C)+CHR$(3): ETX$=CHR$(3)+
CHR$(0): ER$=CHR$(2)+CHR$(&H14)+CHR$(3): ALC$=RQ$+CHR$(0)
100 TRANS$="FMOVE 'F0017.TERMPROG.DATA(FM8TCP)' DN"
110 PRINT #2,TRANS$:CHR$(13);: COLOR 7: PRINT: PRINT TRANS$: PRINT: R$=""
120 R$=R$+INPUT$(LOF(1),#1)
130 IF INSTR(R$,ER$)<>0 THEN COLOR 4: PRINT MID$(R$,5);: GOTO 40
140 IF INSTR(R$,ALC$)=0 THEN GOTO 120
150 OPEN "O", #3, "0:TCP/V4L4": COLOR 4
160 PRINT #2,RQ$:CHR$(13);: R$=""
170 R$=R$+INPUT$(LOF(1),#1)
180 IF INSTR(R$,EDRD$)<>0 THEN GOTO 220
190 IF RIGHT$(R$,2)<>ETX$ THEN GOTO 170
200 R$=MID$(R$,2,LEN(R$)-3)
210 PRINT #3,R$: PRINT R$: GOTO 160
220 COLOR 4: R$=MID$(R$,5)
230 PRINT R$;: IF INSTR(R$,CHR$(0))<>0 THEN CLOSE #3: GOTO 40
240 R$=INPUT$(LOF(1),#1): GOTO 230
250 POKE &HFD07,&H36: CLOSE #3: RESUME 40
260 PRINT #2,ER$;
270 EXEC &HD748: PRINT #2,CHR$(13);: RETURN 40

```

(注意)・受信専用プログラムは、カーソルが表示されない。

- ・入力ミスの修正は BS キーのみである。
- ・アテンションは PF1
- ・受信の強制終了は PF2

9. おわりに

高速の双方向ファイル転送コマンド(!TR)による送受信は、パソコン用のファイル転送コマンド(付録2を参照)によって行っている。現時点での!TRコマンドと!Fコマンドによるファイル転送速度は、正確な計測値はないが、1200bpsの場合、!TRコマンドの方が送信で約2倍、受信で約3倍高速化されている。ただし、この形式によるデータ送信(上り)では、ASCIIの一部(“[“や”]”)などのコード変換が正常に動作しない。これについては、後日ホスト側での変換プログラムを用意する予定である。

なお、今後のレベルアップとして、オリジナル版から削除したプリンターへのエコーバック機能を何とか実現したいと考えている。

端末制御プログラムは▼F0017.TERMPROG.DATA(FM8TCP)▼というデータセット名で、常に最新のものを保存しているので、必要な方は利用して頂きたい。

参考文献

1. 平松敏祐 「マイコンと大型機を結合する端末制御プログラムTCP/C—FM-8によるケース・スタディー」 名古屋大学大型計算機センター研究開発部研究報告, 8, 19-43.

本文では引用していないが下記の文献も参考になる。

2. FUJITSU MICRO 8 F-BASIC 文法書.
3. FUJITSU MICRO 8 ユーザーズマニュアル システム解説.
4. 加瀬 清 「6809 ハンドブック」 アスキー出版, 1982.
5. 野田朋芳 吉里真二 「FM-8のBIOSの使い方とマシン語開発ツールへの応用」 インターフェース, 57, 1982.
6. 野田朋芳 入江英夫 「最新FM-8 技術情報」 インターフェース, 61, 1982.

付録 1.

```

10 '=====
20 '- Terminal Control Program in Color (TCP/C) for HOST & FM-8
30 '- coded by T. Hiramatsu (Nagoya Univ.) on 30/03/82 (V03-L10)
40 '- revised by K. Irie (Kyushu Univ.) on 21/06/82 (V04-L01)
50 '- on 01/09/82 (V04-L03)
60 '- on 07/10/82 (V04-L04)
70 '=====
80 CLEAR 1500,&H6EFF
90 '---- MACHINE LANGUAGE (COMMAND GENERATOR) -- ADDR(6F00--6F31) ----
100 READ A$:J=&H6F00:FOR I=1 TO 100 STEP 2:POKE J,VAL("&H"+MID$(A$,I,2)):J=J+1:N
EXT I
110 DATA AE0210AE84AE02A68481FF26034FA7843410AE0130865FCE0050A6A081192303A7805C3
35F1183000026EF3510EB84E78439
120 '---- MACHINE LANGUAGE (INSERT MODE LED ON/OFF) --- ADDR(6F40--6F61) ----
130 READ B$:J=&H6F40:FOR I=1 TO 68 STEP 2:POKE J,VAL("&H"+MID$(B$,I,2)):J=J+1:NE
XT I
140 DATA 8E6F47BDF2D83910006F4F0013000000003F59414D41543484993D38490B7D40D39
150 COLOR ,0
160 WIDTH 80,25:CONSOLE 0,25,0,0:KEY(1) ON: KEY(10) ON
170 OPEN"I",#1,"COM0:(S7E1)": OPEN"O",#2,"COM0:(S7E1)"
180 DEFINT A-Z
190 DEF USR0=&H6F00 '---- ML ROUTINE ENTRY POINT=&H6F00 ----
200 XMIN=0:YMIN=0:XMAX=79:YMAX=24:X=XMIN:Y=YMIN:YSTP=1:LN3=0:AD=0:N=0:CD=0:INSW=
0:W=0:W1=0:CD1=0:RESPX=0:RESPY=0:BITON=0:D=0:L=0:TRL=0:I=0:J=0
210 DIM SCT(0):DIM KIN(0):DIM WKT(79):DIM SPT(0):DIM DSPCR(39):DIM ADR(1)
220 SCT(0)=0:KIN(0)=0:SPT(0)=0:ADR(0)=0:ADR(1)=0
230 FOR I=0 TO 39:DSPCR(I)=0:NEXT I
240 FOR I=0 TO 79:WKT(I)=0:NEXT I
250 BELL$=CHR$(7)+CHR$(7)+CHR$(0):EOD$="KEQ52802I END OF DATA":CRLF$=CHR$(13)+
CHR$(10)+CHR$(0)
260 NUL$=CHR$(10)+CHR$(0)
270 RD$="READY "+CRLF$: ED$="E"+CRLF$: IN$="INPUT"+CRLF$
280 RQ$=CHR$(2)+CHR$(&H1E)+CHR$(3):EDRD$=CHR$(2)+CHR$(&H1C)+CHR$(3)
290 ER$=CHR$(2)+CHR$(&H14)+CHR$(3):ALC$=RQ$+CHR$(0)
300 CS$="":CMP$="":WS$="":FD$="":DSN$="":RS$="":NUM$="":R3$="":R$="":TRANS$=""
310 W1=VARPTR(SCT(0)):W=VARPTR(KIN(0))
320 ADR(0)=VARPTR(WKT(0)):ADR(1)=VARPTR(CS$):AD=VARPTR(ADR(0))
330 ON KEY(1) GOSUB 1620:ON KEY(10) GOSUB 3260:EXEC &HD748
340 COM(0) OFF:ON COM(0) GOSUB 600
350 FOR I=1 TO 20:GOSUB 2710:NEXT I 'Connecting message throw away
360 LOCATE 0,0
370 BEEP1: COLOR 5: PRINT"* Start of TCP/C (V04-L04) *":BEEP0
380 COLOR 4: PRINT"*** PLEASE LOGON ***";:COLOR 2:PRINT" "
390 ON ERROR GOTO 2780
400 '=====
410 '==== Screen editor (Input from keyboard) section =====
420 '=====
430 CS$="": N=0:TRL=0:LN3=0:CMP$=STRING$(25," ")
440 LOCATE 0,CSRLLIN+1
450 X=0:Y=CSRLLIN
460 '----- Corsor blinking -----
470 GET@A (X,Y)-(X,Y),SCT
480 POKE W1,(PEEK(W1) OR &H08):PUT @A(X,Y)-(X,Y),SCT
490 COM(0) ON
500 I=0
510 I=I+1:IF I>20 THEN GOTO 520 ELSE A$=INKEY$:IF A$<>" " THEN GOTO 650 ELSE GOTO
510
520 COM(0) STOP
530 GET@A (X,Y)-(X,Y),SCT
540 POKE W1,(PEEK(W1) AND &HF7):PUT @A(X,Y)-(X,Y),SCT
550 COM(0) ON
560 I=0

```

```

570 I=I+1:IF I>20 THEN GOTO 580 ELSE A$=INKEY$:IF A$<>" " THEN GOTO 670 ELSE GOTO
570
580 COM(0) STOP
590 GOTO 470
600 GET@A (X,Y)-(X,Y),SCT
610 POKE W1,(PEEK(W1) AND &HF7):PUT @A(X,Y)-(X,Y),SCT
620 GOSUB 2710
630 X=POS(0):Y=CSRLIN:RESPX=POS(0):RESPY=CSRLIN
640 RETURN
650 POKE W1,(PEEK(W1) AND &HF7):PUT @A(X,Y)-(X,Y),SCT
660 '-----
670 CD=ASC(A$):IF CD=5 THEN GOTO 1440 ELSE IF CD<32 GOTO 960
680 IF CD=127 GOTO 1400
690 IF INSW=0 THEN GOTO 790
700 '----- Insert process -----
710 IF X=>XMAX THEN GOTO 1450
720 GET @(XMAX,Y)-(XMAX,Y),SCT
730 IF PEEK(W1)>=32 THEN GOTO 1450
740 GET @A(X,Y)-(XMAX-1,Y),WKT
750 POKE W,CD:PUT @(X,Y)-(X,Y),KIN,2:MID$(CMP$,Y+1,1)="1"
760 PUT @A(X+1,Y)-(XMAX,Y),WKT
770 GOTO 840
780 '----- Put key in charactor -----
790 POKE W,CD:PUT @(X,Y)-(X,Y),KIN,2:MID$(CMP$,Y+1,1)="1"
800 GOTO 840
810 '----- [ -> ] -----
820 X=X+1:IF X>XMAX THEN X=XMIN
830 GOTO 470
840 X=X+1:IF X>XMAX GOTO 860
850 GOTO 470
860 X=XMIN
870 '----- [ down ] -----
880 Y=Y+YSTP:IF Y>YMAX THEN Y=YMIN
890 GOTO 470
900 '----- [ <- ] -----
910 X=X-1:IF X<XMIN THEN X=XMAX
920 GOTO 470
930 '----- [ up ] -----
940 Y=Y-YSTP:IF Y<YMIN THEN Y=YMAX
950 GOTO 470
960 CD1=CD-7:IF CD1<1 GOTO 1450
970 ON CD1 GOTO 1010,1080,1450,1110,1320,1150,1450,1450,1450,1340,1370
980 CD1=CD-27:IF CD1<1 GOTO 1450
990 ON CD1 GOTO 820,910,940,880
1000 '----- [ BS ] -----
1010 IF X=XMIN GOTO 1450
1020 GET @A(X,Y)-(XMAX,Y),WKT
1030 PUT @A(X-1,Y)-(XMAX-1,Y),WKT
1040 PUT @(XMAX,Y)-(XMAX,Y),SPT
1050 X=X-1:MID$(CMP$,Y+1,1)="1"
1060 GOTO 470
1070 '----- [ TAB ] -----
1080 X=XMIN
1090 GOTO 880
1100 '----- [ HOME ] -----
1110 IF X=XMIN AND Y=YMIN THEN Y=YMAX :GOTO 470
1120 IF X=XMIN THEN Y=YMIN:GOTO 470
1130 X=XMIN:GOTO 470
1140 '----- [ RETURN ] -----
1150 INSW=0:POKE &H6F5E,&HB7:EXEC &H6F40:CSS$=STRING$(255," ")
1160 IF RESPX=0 THEN GOTO 1210
1170 GET @(RESPX,RESPY)-(XMAX,RESPY),WKT

```

```

1180 AD=USRO(AD)
1190 GOTO 1290
1200 '--- Multi line ---
1210 BITON=INSTR(CMP$, "1")-1
1220 IF BITON<0 OR LN3=>3 THEN GOTO 1280
1230 LN3=LN3+1:MID$(CMP$, BITON+1, 1)=" "
1240 GET @(XMIN, BITON)-(XMAX, BITON), WKT
1250 AD=USRO(AD)
1260 PUT @(XMIN, BITON)-(XMAX, BITON), WKT, 2
1270 GOTO 1210
1280 IF LEN(CS$)=255 THEN CS$=""
1290 COM(0) OFF
1300 GOTO 1500
1310 '----- [ CLEAR ] -----
1320 CLS:X=0:Y=0:CMPS$="":CMPS$=STRING$(25, " "):GOTO 470
1330 '----- [ DUP ] -----
1340 IF Y=YMIN GOTO 1450
1350 GET @A(X, Y-YSTP)-(X, Y-YSTP), SCT:PUT @A(X, Y)-(X, Y), SCT:MID$(CMPS$, Y+1, 1)="1":
GOTO 840
1360 '----- [ INS ] -----
1370 IF INSW<>0 THEN INSW=0:POKE &H6F5E, &HB7 ELSE INSW=1:POKE &H6F5E, &HB6
1380 EXEC &H6F40:GOTO 470 '&H6F40=Insert Mode LED on/off
1390 '----- [ DEL ] -----
1400 IF X>=XMAX GOTO 1420
1410 GET @A(X+1, Y)-(XMAX, Y), WKT:PUT @A(X, Y)-(XMAX-1, Y), WKT
1420 PUT @(XMAX, Y)-(XMAX, Y), SPT:MID$(CMPS$, Y+1, 1)="1":GOTO 470
1430 '----- [ EL ] -----
1440 PUT @A(X, Y)-(XMAX, Y), DSEPCR, 2:MID$(CMPS$, Y+1, 1)="1":GOTO 470
1450 BEEP 1:FOR I=1 TO 50:NEXT I:BEEP 0
1460 GOTO 470
1470 '=====
1480 '===== Command select section =====
1490 '=====
1500 IF LEFT$(CS$, 1)<>"!" THEN PRINT #2, CS$:CHR$(13);:GOTO 430
1510 IF CS$="!" THEN GOSUB 1620
1520 IF CS$="!TR" OR CS$="!tr" THEN 2790
1530 IF CS$="!C" OR CS$="!c" THEN 1720
1540 IF CS$="!E" OR CS$="!e" THEN 1740
1550 IF CS$="!F" OR CS$="!f" THEN 1930
1560 IF CS$="!L" OR CS$="!l" THEN 1760
1570 IF CS$="!M" OR CS$="!m" THEN MOTOR: GOTO 430
1580 IF CS$="!S" OR CS$="!s" THEN 1780
1590 IF CS$="!?" OR CS$="!H" OR CS$="!h" THEN GOSUB 1860: GOTO 430
1600 COLOR2: PRINT "TCP/C COMMAND ERROR": GOSUB 2670: GOTO 430
1610 '----- Interrupt -----
1620 IF TR1=1 THEN PRINT #2, ER$:EXEC &HD748:CLOSE #3:RETURN 430
1630 IF TR1=2 THEN PRINT #2, EDRD$:EXEC &HD748:CLOSE #3:RETURN 430
1640 IF N=0 THEN GOSUB 1700: CLOSE #3: RETURN 430
1650 ON N GOTO 1660, 1660, 1670, 1680, 1680
1660 GOSUB 1700: RETURN 2330
1670 GOSUB 1690: RETURN 2510
1680 GOSUB 1690: RETURN 2580
1690 COLOR 5: PRINT "Break !I": RETURN
1700 EXEC &HD748: GOSUB 2710: GOSUB 1690: RETURN
1710 '----- CRT ---> Printer -----
1720 PRINT CHR$(30);:HARDC:COLOR5:PRINT"End of Hardcopy":GOSUB2670:GOTO 430
1730 '----- LOGOFF command -----
1740 PRINT #2, "LOGOFF";CHR$(13);: GOTO 430
1750 '----- LOGON command -----
1760 CS$="LOGON TSS F0000/##### ":WS$=RD$:GOSUB2620:GOSUB1810:GOTO 430
1770 '----- Stop TCP -----
1780 PRINT "*" STOP of TCP *": PRINT "type in 'CONT' to restart";

```

```

1790 COLOR 7: STOP: GOSUB 2670: COLOR 5: PRINT "* Restart of TCP/C *":GOTO 430
1800 '----'
1810 CS$="TERM NOTIMEOUT LINESIZE(255) ": GOTO 1840
1820 CS$="TERM LINES(1)":GOTO1840
1830 CS$="TERM NOLINES"
1840 WS$=RD$: COLOR 7: PRINT CS$: GOSUB 2620:RETURN
1850 '----- [ !? ] or [ !H ] -----
1860 COLOR 5: PRINT: PRINT "* List of TCP/C command *": PRINT
1870 PRINT "!C---Hardcopy, !E---Logoff, !L-----Logon, !F-----File"
1880 PRINT "!M---CT Motor, !S---Stop, !?(!H)---Help, !TR----Transfer"
1890 PRINT "PF1---BREAK "
1900 PRINT "PF10--Program End ": GOSUB 2670:RETURN
1910 '----'
1920 '----- [ !F ] -----
1930 PRINT: COLOR 5
1940 PRINT "-----"
1950 PRINT " 1 --- Receive File from HOST (Open FM-8 File, New Mode)"
1960 PRINT " 2 --- Receive File from HOST (Open FM-8 File, Append Mode)"
1970 PRINT " 3 --- Send FM-8 File to HOST"
1980 PRINT " 4 --- Display FM-8 File"
1990 PRINT " 5 --- Print FM-8 File"
2000 PRINT " 6 --- Close FM-8 File / End of !F Command"
2010 PRINT "-----"
2020 PRINT "* which case (1,2,--,6) ? =";: COLOR 7: INPUT" ",N: COLOR 5
2030 '----'
2040 IF N>6 OR N<0 THEN COLOR 5:PRINT: PRINT "* End of !F Command *":GOTO 2600
2050 COLOR 5: PRINT "* Device Type";
2060 PRINT" (1--CT, 2--FPD0, 3--FPD1, 4--BUB0, 5--BUB1) ? =";:COLOR 7:INPUT" ",D
2070 IF D=1 THEN FD$="CAS0:": GOTO 2100: ELSE IF D=2 THEN FD$="0:":GOTO 2100
2080 IF D=3 THEN FD$="1:": GOTO 2100: ELSE IF D=4 THEN FD$="BUB0:": GOTO 2100
2090 IF D=5 THEN FD$="BUB1:": GOTO 2100: ELSE N=6: GOTO 2040
2100 COLOR 5: PRINT"* file name of FM-8 ? =";: COLOR 7:INPUT" ",DSN$:FD$=FD$+DSN
$
2110 ON N GOTO 2160,2230,2360,2550,2550
2120 '=====
2130 '===== File transfer section =====
2140 '=====
2150 '----- Host ---> FM-8 (new mode) -----
2160 COLOR5:PRINT"* Open File (;FD$;)-- New Mode -- *":OPEN"O",#3,FD$:GOSUB 18
20
2170 COLOR5:PRINT"* 'dsname' of HOST ? =";: COLOR7: INPUT" ",DSN$: PRINT #3,"
2180 CS$="LIST "+DSN$: PRINT CS$: PRINT: PRINT #2,CS$:CHR$(13);: RS$=""
2190 RS$=RS$+INPUT$(LOF(1),#1):IF RIGHT$(RS$,3)<>BELL$ GOTO2190 ELSE L=LEN(RS$)-
14
2200 COLOR4:CS$=MID$(RS$,2,L):PRINT CS$:PRINT #2,CHR$(13);
2210 GOTO 2270
2220 '----- Host ---> FM-8 (append mode) --
2230 COLOR 5: PRINT "* Prepare Floppy Disk (Append Mode) *":GOSUB1820
2240 IF D <> 2 AND D <> 3 THEN COLOR 2: PRINT "* INVALID DEVICE *": GOTO 1930
2250 OPEN "A",#3,FD$: GOTO 2170
2260 '----- Host ---> FM-8 (transfer) -----
2270 R$=INPUT$(LOF(1),#1):IF R$=""THEN 2270 ELSE R$=RIGHT$(R$,LEN(R$)-1):RS$=""
2280 R$=R$+INPUT$(LOF(1),#1): L=LEN(R$)-13: IF L < 0 GOTO 2280
2290 RS$=R$+LEFT$(R$,L): R$=RIGHT$(R$,13): IF RIGHT$(R$,3)<>BELL$ GOTO 2280
2300 PRINT RS$:IF RS$=EOD$ THEN 2320 ELSE PRINT #3,RS$
2310 PRINT #2,CHR$(13);:GOTO 2270
2320 CS$="": WS$=RD$: GOSUB 2620
2330 CLOSE #3: COLOR5: CS$="End of Receive File": PRINT CS$
2340 GOSUB 1830: GOTO 430
2350 '----- FM-8 ---> Host -----
2360 OPEN "I",#3,FD$: GOSUB 1830

```

```

2370 COLOR5:PRINT* 'dsname' of HOST ? =";COLOR7:INPUT" ",DSN$:COLOR5:PRINT"rem
ove line number ? Y OR N ";:COLOR7:INPUT" ",NUM$:CS$="E "+DSN$
2380 CS$=CS$+" NEW NONUM":COLOR7: PRINT CS$: WS$=IN$: GOSUB 2620
2390 IF NUM$="Y" THEN GOTO 2450
2400 IF EOF(3) GOTO 2510 ELSE LINE INPUT #3,CS$:R3$="": IF CS$="" GOTO 2400
2410 PRINT #2,CS$;CHR$(13);:COLOR6:PRINT CS$
2420 R$=INPUT$(LOF(1),#1): R3$=R3$+R$:L=LEN(R3$):PRINT R$;: IF L<2 GOTO 2420
2430 IF L=2 AND R3$=NUL$ GOTO 2400 ELSE COLOR4: PRINT R$;
2440 GOSUB 2710: GOTO 2410
2450 IF EOF(3) GOTO 2510 ELSE LINE INPUT #3,CS$:R3$="": IF CS$="" GOTO 2450
2460 CS$=MID$(CS$,INSTR(CS$,"")+1):IF CS$="" GOTO 2450
2470 PRINT #2,CS$;CHR$(13);:COLOR6:PRINT CS$
2480 R$=INPUT$(LOF(1),#1): R3$=R3$+R$:L=LEN(R3$):PRINT R$;: IF L<2 GOTO 2480
2490 IF L=2 AND R3$=NUL$ GOTO 2450 ELSE COLOR4: PRINT R$;
2500 GOSUB 2710: GOTO 2470
2510 COLOR 5:PRINT:CS$="* End of Send File *":PRINT CS$: CLOSE #3
2520 CS$="": WS$=ED$:GOSUB 2620: PRINT #2,"SAVE";CHR$(13);:COLOR 7:PRINT "SAVE"
2530 GOTO 430
2540 '----- Display (FM-8 file) section ---
2550 OPEN "I",#3,FD$: PRINT: IF N=5 THEN CLOSE #4: OPEN "O",#4,"LPT0:"
2560 IF EOF(3) GOTO 2580 ELSE LINE INPUT #3,CS$: IF CS$="" THEN 2560
2570 COLOR 6: PRINT CS$: IF N <> 5 GOTO 2560 ELSE PRINT #4,CS$: GOTO 2560
2580 COLOR5: PRINT: PRINT "End of Display/Print"
2590 '----'
2600 GOSUB 2670: CLOSE #3:GOSUB 2710: GOTO 430
2610 '----'
2620 PRINT #2,CS$;CHR$(13);:RS$="":COLOR 4
2630 R$=INPUT$(LOF(1),#1): IF R$="" GOTO 2630 ELSE PRINT R$;:RS$=RS$+R$
2640 IF INSTR(RS$,"KEQ54111A") <> 0 THEN GOSUB 2710:COLOR 7:PRINT CS$:GOTO 2620
2650 IF INSTR(RS$,WS$)=0 THEN 2630 ELSE PRINT:RETURN
2660 '----'
2670 PRINT #2,CHR$(13);
2680 '=====
2690 '===== Display section =====
2700 '=====
2710 COLOR 4
2720 IF LOF(1)=0 THEN RETURN ELSE R$=INPUT$(LOF(1),#1)
2730 L=80-POS(0):IF L<LEN(R$) THEN PRINT LEFT$(R$,L);:R$=MID$(R$,L+1):GOTO 2730
2740 PRINT R$;:GOTO 2720
2750 '=====
2760 '===== Error process section =====
2770 '=====
2780 COLOR 2:PRINT"error code=" ERR;," at (" ERL ")":POKE &HFD07,&H36:RESUME 260
0
2790 '=====
2800 '===== !TR          FMOVE COMMAND =====
2810 '=====
2820 COLOR 5:PRINT:PRINT"          FILE TRANSFER (FMOVE COMMAND)"
2830 PRINT"1          HOST =====> FM-8 "
2840 PRINT"2          FM-8 =====> HOST "
2850 PRINT"3          !TR command end "
2860 PRINT"1 or 2 or 3 ?":COLOR 7:INPUT" ",TR1
2870 IF TR1>2 OR TR1=<0 THEN PRINT #2,CHR$(13);:GOTO 430
2880 COLOR 5:PRINT"FM-8          FILE DISCRIPTOR ?":COLOR 7:INPUT" ",FD$
2890 IF FD$=SPACE$(LEN(FD$)) THEN PRINT #2,CHR$(13);:GOTO 430
2900 COLOR 5:PRINT"HOST          DATA SET NAME ?":COLOR 7:INPUT" ",DSN$
2910 IF DSN$=SPACE$(LEN(DSN$)) THEN PRINT #2,CHR$(13);:GOTO 430
2920 ON TR1 GOTO 2930,3150
2930 'HOST =====> FM-8 (FMOVE COMMAND)
2940 OPEN "O",#3,FD$
2950 TRANS$="FMOVE "+DSN$+" DN"
2960 PRINT #2,TRANS$;CHR$(13);:COLOR 7:PRINT:PRINT TRANS$:PRINT:R$="":COLOR 4

```

```

2970 R$=R$+INPUT$(LOF(1),#1)
2980 IF INSTR(R$,ER$)<>0 THEN PRINT MID$(R$,5);:GOTO 430
2990 IF INSTR(R$,ALC$)=0 THEN GOTO 2970
3000 PRINT#2,RQ$;CHR$(13);:R$="":RS$=""
3010 R$=R$+INPUT$(LOF(1),#1)
3020 IF LEN(R$)<5 THEN GOTO 3010
3030 IF INSTR(R$,EDRD$)<>0 THEN GOTO 3120
3040 R$=MID$(R$,INSTR(R$,CHR$(2))+1)
3050 IF RIGHT$(R$,1)=CHR$(0) THEN R$=LEFT$(R$,LEN(R$)-2):GOTO 3110
3060 IF RIGHT$(R$,1)=CHR$(3) THEN R$=LEFT$(R$,LEN(R$)-1) ELSE GOTO 3080
3070 IF INPUT$(LOF(1),#1)=CHR$(0) THEN GOTO 3110 ELSE GOTO 3070
3080 RS$=RS$+INPUT$(LOF(1),#1)
3090 IF RIGHT$(RS$,1)<>CHR$(0) THEN GOTO 3080
3100 R$=R$+LEFT$(RS$,LEN(RS$)-2)
3110 PRINT #3,R$:PRINT R$:GOTO 3000
3120 COLOR 4:R$=MID$(R$,5)
3130 PRINT R$;:IF INSTR(R$,CHR$(0))<>0 THEN CLOSE #3:GOTO 430
3140 R$=INPUT$(LOF(1),#1):GOTO 3130
3150 'FM-8 =====> HOST (FMOVE COMMAND)
3160 OPEN "I",#3,FD$
3170 TRANS$="FMOVE "+DSN$+" UP"
3180 PRINT #2,TRANS$;CHR$(13);:COLOR 7:PRINT:PRINT TRANS$:PRINT:R$="":COLOR 6
3190 R$=R$+INPUT$(LOF(1),#1)
3200 IF INSTR(R$,ER$)<> 0 THEN COLOR 4:PRINT MID$(R$,5);:GOTO 430
3210 IF INSTR(R$,ALC$)=0 THEN GOTO 3190
3220 IF EOF(3)<>0 THEN GOTO 3250
3230 LINE INPUT #3,RS$:IF RS$="" THEN GOTO 3220
3240 PRINT #2,CHR$(2);RS$;CHR$(3);:PRINT RS$:R$=INPUT$(LOF(1),#1):GOTO 3220
3250 CLOSE #3:PRINT #2,EDRD$;:GOTO 430
3260 '===== TCP/C End =====
3270 CLOSE:COLOR5:PRINT:PRINT"* End of TCP/C *":FOR I=1 TO 10:KEY(I) OFF:NEXT I:
COLOR7:END

```

付録2.

ファイル転送のためのコマンド

コマンド名	オペランド
FMOVE	データセット名 $\left[\begin{array}{c} \text{UP} \\ \text{DN} \end{array} \right]$

オペランド

データセット名：ホストのデータセット名を指定する。区分データセットの場合は、メンバ名まで指定する。

UP : 端末からホストへのファイル転送。

DN : ホストから端末へのファイル転送。

使用上の注意

UPを指定する場合は以下のことに注意する。


- データセットが、すでに存在すればリプレースされる。
- 新しくデータセットを作成する場合、データセットタイプがTEXTとCLISTはRECFM = VB, LRECL = 255, BLKSIZE = 3120, それ以外は, RECFM = FB, LRECL = 80, BLKSIZE = 3120で作成される。

(i) 伝送手順


UPおよびDNにおける伝送手順を以下に示す。なお、下記において、 で囲んだ2桁の文字は1バイトを16進数で表現したものである。

正常時の伝送手順(上り)

端 末	回 線 上	ホ ス ト
コマンドの送出	$\boxed{\text{FMOVE}} _ \boxed{\text{データセット名}} _ \boxed{\text{UP}} \text{ } \boxed{\text{C}_R}$ →	レコードの要求
	← $\boxed{00} \boxed{02} \boxed{1E} \boxed{03} \boxed{00}$	
レコードの送出	$\boxed{02} _ \boxed{1 \text{レコード}} _ \boxed{03}$ →	後続レコードの要求
	← $\boxed{00}$	
レコードの送出	$\boxed{02} _ \boxed{1 \text{レコード}} _ \boxed{03}$ →	



エラー発生時の処理手順



正常時の伝送手順(下り)

端 末	回 線 上	ホ ス ト
コマンドの送出	FMOVE_データセット名_DN C _R →	
	← 00 02 1E 03 00	初期設定完了の通知
レコードの要求	02 1E 03 →	
	← 00 0A 02 1レコード 03 00	レコードの送出
レコードの要求	02 1E 03 →	
	← 00 02 1レコード 03 00	レコードの送出
レコードの要求	02 1E 03 →	
	⋮	
	02 1E 03 →	
	← 00 02 1C 03	EOFの送出
	← _PFM3000I_..... C _R L _F	終了メッセージの送出
	← _TOTAL_RECORD..... C _R L _F	送信情報の送出
	← READY_ C _R L _F 00	モードメッセージの送出

エラー発生時の処理手順

	← 0A 02 14 03	エラー発生の通知
	← _ エラーメッセージ C _R L _F	エラーメッセージの送出
	⋮	
	← READY_ C _R L _F 00	モードメッセージの送出